A robust teaching-learning platform dedicated for you to gain the industrial skills and knowledge by, making your learning journey more valuable.

Financial Analysis Project in Excel (Documentation)

Table of Contents

Introduction

The purpose of this project is to provide fundamental insights into how financial functions and terminologies work in Excel. It is essential to thoroughly understand the provided documentation and terms to efficiently utilize Excel's financial functions such as PV, NPV, XNPV, IRR, MIRR, and XIRR. This project involves analyzing a prepared dataset and applying these financial functions to derive meaningful financial insights.

Project Scope and Objectives

Scope

- Utilize Excel to perform financial analysis.
- Apply various financial functions to solve real-world financial problems.
- Create a summary report based on the analysis conducted.

Objectives

- Understand and use Excel functions for financial analysis.
- Calculate present value, net present value, and internal rate of return for different scenarios.
- Determine the effective interest rate, loan term, and EMI for given financial data.
- Generate a detailed summary report of insights and analysis.

Key Financial Concepts and Functions

Annuity

An annuity is a series of constant cash payments made over a continuous period, such as retirement savings, insurance payments, or mortgage payments. In annuity functions:

- A positive number represents cash received.
- A negative number represents cash paid out.

Present Value (PV)

The present value is the total amount that a series of future payments is worth now.

Excel Function: PV

=PV(rate, nper, pmt, [fv], [type])

- rate: Interest rate per period.
- nper: Total number of payment periods.
- pmt: Payment made each period.
- fv: Future value (optional).
- type: Payment type (0 for end of period, 1 for beginning of period).

Example

- Price of refrigerator: 32,000

- Annual interest rate: 13%
- Number of years: 8
- Yearly payment: 6,000

Calculate PV with payments at the end of the year:

=PV(13%, 8, -6000)

Net Present Value (NPV)

NPV calculates the net present value of an investment based on a discount rate and a series of future payments and incomes.

Excel Function: NPV

=NPV(rate, value1, [value2], ...)

- rate: Discount rate.
- value1, value2, ...: Cash flows.

Example

Investment with different cash flows:

=NPV(10%, -1000, 300, 400, 500, 600)

XNPV

XNPV calculates the net present value for a schedule of cash flows that are not necessarily periodic.

Excel Function: XNPV

=XNPV(rate, values, dates)

- rate: Discount rate.
- values: Cash flows.
- dates: Corresponding dates of the cash flows.

Equated Monthly Installment (EMI)

An EMI is a fixed payment amount made by a borrower to a lender at a specified date each calendar month.

Excel Function: PMT

=PMT(rate, nper, pv, [fv], [type])

- rate: Interest rate per period.
- nper: Total number of payment periods.
- pv: Present value or loan amount.
- fv: Future value (optional).
- type: Payment type (0 for end of period, 1 for beginning of period).

Example

- Loan amount: 5,000,000
- Annual interest rate: 11.5%
- Loan term: 25 years

Calculate EMI:

=PMT(11.5%/12, 25*12, -5000000, 0, 1)

Interest and Principal Components of EMI

To calculate the interest and principal parts of the EMI:

- **Interest Component: IPMT**

=IPMT(rate, per, nper, pv, [fv], [type])

- **Principal Component: PPMT**

=PPMT(rate, per, nper, pv, [fv], [type])

Calculating Interest Rate

To find the interest rate required to pay back a loan: **Excel Function: RATE**

=RATE(nper, pmt, pv, [fv], [type], [guess])

Example

- Loan amount: 100,000
- Maximum monthly payment: 12,000
- Loan term: 15 months

Calculate interest rate:

=RATE(15, -12000, 100000)

Calculating Loan Term

To determine the number of payments required to clear a loan:
Excel Function: NPER

=NPER(rate, pmt, pv, [fv], [type])

Example

- Loan amount: 100,000
- Monthly payment: 15,000
- Interest rate: 10%

Calculate loan term:

=NPER(10%/12, -15000, 100000)

Internal Rate of Return (IRR)

IRR is the rate of interest at which NPV is zero.

Excel Function: IRR

=IRR(values, [guess])

- values: Cash flows.
- guess: Initial guess (optional).

Example

Calculate IRR for given cash flows:

=IRR(A1:A5)

XIRR

To calculate the IRR for irregularly spaced cash flows: **Excel Function: XIRR**

=XIRR(values, dates, [guess])

Modified IRR (MIRR)

MIRR takes into account different finance and reinvestment rates.

Excel Function: MIRR

=MIRR(values, finance_rate, reinvest_rate)

Summary Report

Insights and Analysis

- **Annuity**: Understood the concept of constant cash payments over time.
- **Present Value (PV)**: Learned to calculate the current worth of future payments.
- **Net Present Value (NPV) and XNPV**: Compared different investments based on future cash flows.
- **EMI**: Determined the monthly installment for a loan and its components.
- **Interest Rate and Loan Term**: Calculated the effective interest rate and the duration required to repay a loan.
- **Internal Rate of Return (IRR) and Modified IRR (MIRR)**: Assessed the profitability of investments considering different cash flow timings and rates.

The project provided a comprehensive understanding of financial functions in Excel, enabling effective financial analysis and decision-making.

Implementations and Examples

Example

Suppose you are buying a refrigerator. The salesperson tells you that the price of the refrigerator is 32000, but you have an option to pay out the amount in 8 years with an interest rate of 13% per annum and yearly payments of 6000. You also have an option to make the payments either at the beginning or end of each year.

You want to know which of these options is beneficial for you.

You can use Excel function PV −

PV (rate, nper, pmt, [fv], [type])

To calculate present value with payments at the end of each year, omit type or specify 0 for type.

To calculate present value with payments at the end of each year, specify 1 for type.

	A	B	C
1			
2		**Price**	32000
3		**Interest Rate**	0.13
4		**No. of Payments**	8
5		**Payment**	-6000
6		**Payment at end of each year**	
7		**PV**	=PV(C3,C4,C5)
8			
9		**Price**	32000
10		**Interest Rate**	0.13
11		**No. of Payments**	8
12		**Payment**	-6000
13		**Payment at beginning of each year**	
14		**PV**	=PV(C10,C11,C12,,1)

You will get the following results −



	A	B	C
1			
2		Price	32000
3		Interest Rate	13%
4		No. of Payments	8
5		Payment	-6000
6		Payment at end of each year	
7		PV	28,793
8			
9		Price	32000
10		Interest Rate	13%
11		No. of Payments	8
12		Payment	-6000
13		Payment at beginning of each year	
14		PV	32,536

Therefore,

If you make the payment now, you need to pay 32,000 of present value.

If you opt for yearly payments with payment at the end of the year, you need to pay 28, 793 of present value.

If you opt for yearly payments with payment at the end of the year, you need to pay 32,536 of present value.

You can clearly see that option 2 is beneficial for you.

Suppose, you want to take a home loan of 5000000 with an annual interest rate of 11.5% and the term of the loan for 25 years. You can find your EMI as follows −

Calculate interest rate per month (Interest Rate per Annum/12)

Calculate number of monthly payments (No. of years * 12)

Use PMT function to calculate EMI



	A	B	C
1			
2		**Rate per Annum**	0.12
3		**Rate per Month**	=C2/12
4		**Term**	25
5		**No. of Monthly Payments**	=C4*12
6		**Loan Amount (PV)**	5000000
7		**FV**	0
8		**type**	1
9		**EMI**	=PMT(C3,C5,C6,C7,C8)

As you observe,

Present Value (PV) is the loan amount.

Future Value (FV) is 0 as at the end of the term the loan amount should be 0.

Type is 1 as the EMIs are paid at the beginning of each month.

You will get the following results −



	A	B	C
1			
2		Rate per Annum	12%
3		Rate per Month	0.01
4		Term	25
5		No. of Monthly Payments	300
6		Loan Amount (PV)	5000000
7		FV	0
8		type	1
9		EMI	($52,139.81)

Monthly Payment of Principal and Interest on a Loan

EMI includes both-interest and a part payment of principal. As the time increases, these two components of EMI will vary, reducing the balance.

To get

The interest part of your monthly payments, you can use the Excel IPMT function.

The payment of principal part of your monthly payments, you can use the Excel PPMT function.

For example, if you have taken a loan of 1,000,000 for a term of 8 months at the rate of 16% per annum. You can get values for

the EMI, the decreasing interest amounts, the increasing payment of principal amounts and the diminishing loan balance over the 8 months. At the end of 8 months, loan balance will be 0.

Follow the procedure given below.



	A	B	C
1			
2		**Rate per Annum**	0.16
3		**Rate per Month**	=C2/12
4		**No. of Monthly Payments**	8
5		**Loan Amount (PV)**	100000
6		**FV**	0
7		**type**	0
8		**EMI**	=PMT(C3,C4,C5,C6,C7)

This results in an EMI of Rs. 13261.59.



	A	B	C	D
1				
2		Rate per Annum	16%	
3		Rate per Month	0.01	
4		No. of Monthly Payments	8	
5		Loan Amount (PV)	100000	
6		FV	0	
7		type	0	
8		EMI	(13,261.59)	

Step 2 − Next calculate the interest and principal parts of the EMI for the 8 months as shown below.

	A	B	C	D	E	F	G
1							
2			Rate per Month	0.013			
3			No. of Monthly Payments	8			
4			Loan Amount (PV)	100000			
5			FV	0			
6			type	0			
7			EMI	-13242.27			
8							
9		Month	Beginning Balance	EMI	Interest	Principal	Ending Balance
10		1	=D4	=-$D$7	=-IPMT($D$2,B10,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B10,$D$3,$D$4,,$D$6)	=C10-F10
11		2	=G10	=-$D$7	=-IPMT($D$2,B11,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B11,$D$3,$D$4,,$D$6)	=C11-F11
12		3	=G11	=-$D$7	=-IPMT($D$2,B12,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B12,$D$3,$D$4,,$D$6)	=C12-F12
13		4	=G12	=-$D$7	=-IPMT($D$2,B13,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B13,$D$3,$D$4,,$D$6)	=C13-F13
14		5	=G13	=-$D$7	=-IPMT($D$2,B14,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B14,$D$3,$D$4,,$D$6)	=C14-F14
15		6	=G14	=-$D$7	=-IPMT($D$2,B15,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B15,$D$3,$D$4,,$D$6)	=C15-F15
16		7	=G15	=-$D$7	=-IPMT($D$2,B16,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B16,$D$3,$D$4,,$D$6)	=C16-F16
17		8	=G16	=-$D$7	=-IPMT($D$2,B17,$D$3,$D$4,,$D$6)	=-PPMT($D$2,B17,$D$3,$D$4,,$D$6)	=C17-F17

You will get the following results.

9	Month	Beginning Balance	EMI	Interest	Principal	Ending Balance
10	1	100000	13,242.27	1,300.00	11942.27	88057.73
11	2	88057.73	13,242.27	1,144.75	12097.52	75960.22
12	3	75960.22	13,242.27	987.48	12254.78	63705.43
13	4	63705.43	13,242.27	828.17	12414.10	51291.34
14	5	51291.34	13,242.27	666.79	12575.48	38715.86
15	6	38715.86	13,242.27	503.31	12738.96	25976.89
16	7	25976.89	13,242.27	337.70	12904.57	13072.33
17	8	13072.33	13,242.27	169.94	13072.33	0.00

Interest and Principal paid between two Periods

You can compute the interest and principal paid between two periods, inclusive.

Compute the cumulative interest paid between 2nd and 3rd months using the CUMIPMT function.

Verify the result summing up the interest values for 2nd and 3rd months.

Compute the cumulative principal paid between 2nd and 3rd months using

	A	B	C	D	E	F	G
1							
2			**Rate per Month**	0.013		**Interest paid between 2nd and 3rd Months**	
3			**No. of Monthly Payments**	8		=-CUMIPMT(D2,D3,D4,2,3,D6)	
4			**Loan Amount (PV)**	100000		=E11+E12	
5			**FV**	0		**Principal paid between 2nd and 3rd Months**	
6			**type**	0		=-CUMPRINC(D2,D3,D4,2,3,D6)	
7			**EMI**	-13242.27		=F11+F12	
8							
9		**Month**	**Beginning Balance**	**EMI**	**Interest**	**Principal**	**Ending Balance**
10		1	100000	13242.27	1300	11942.27	88057.73
11		2	88057.73	13242.27	1144.75	12097.52	75960.22
12		3	75960.21	13242.27	987.48	12254.78	63705.43
13		4	63705.43	13242.27	828.17	12414.09	51291.34
14		5	51291.34	13242.27	666.79	12575.48	38715.86
15		6	38715.86	13242.27	503.31	12738.96	25976.89
16		7	25976.89	13242.27	337.69	12904.57	13072.33
17		8	13072.33	13242.27	169.94	13072.33	0

You will get the following results.

	A	B	C	D	E	F	G
1							
2			Rate per Month	0.01		Interest paid between 2nd and 3rd Months	
3			No. of Monthly Payments	8		2,132.23	
4			Loan Amount (PV)	100000		2132.23	
5			FV	0		Principal paid between 2nd and 3rd Months	
6			type	0		24352.30099	
7			EMI	(13242.27)		24352.30	
8							

Month	Beginning Balance	EMI	Interest	Principal	Ending Balance
1	100000.00	13242.27	1300.00	11942.27	88057.73
2	88057.73	13242.27	1144.75	12097.52	75960.22
3	75960.21	13242.27	987.48	12254.78	63705.43
4	63705.43	13242.27	828.17	12414.09	51291.34
5	51291.34	13242.27	666.79	12575.48	38715.86
6	38715.86	13242.27	503.31	12738.96	25976.89
7	25976.89	13242.27	337.69	12904.57	13072.33
8	13072.33	13242.27	169.94	13072.33	0.00

You can see that your calculations match with your verification results.

Calculating Interest Rate

Suppose you take a loan of 100,000 and you want to pay back in 15 months with a maximum monthly payment of 12000. You might want to know the interest rate at which you have to pay.

Find the interest rate with the Excel RATE function −

	A	B	C
1			
2		**Loan Amount**	100000
3		**No. of Monthly Payments**	15
4		**EMI**	-12000
5		**Interest**	=RATE(C3,C4,C2,,0,)

You will get the result as 8%



	A	B	C
1			
2		Loan Amount	100000
3		No. of Monthly Payments	15
4		EMI	-12000
5		Interest	8%

Calculating Term of Loan

Suppose you take a loan of 100,000 at the interest rate 10%. You want a maximum monthly payment of 15,000. You might want to know how long it will take for you to clear the loan.

Find the number of payments with Excel NPER function

	A	B	C
1			
2		Loan Amount	100000
3		Interest	0.1
4		EMI	-15000
5		No. of Monthly Payments	=NPER(C3,C4,C2,,0)

You will get the result as 12 months

	A	B	C
1			
2		Loan Amount	100000
3		Interest	10%
4		EMI	-15000
5		No. of Monthly Payments	12

Decisions on Investments

When you want to make an investment, you compare the different options and choose the one that yields better returns. Net present value is useful in comparing cash flows over a period of time and deciding which one is better. The cash flows can occur at regular, periodical intervals or at irregular intervals.

First, we consider the case of **regular, periodical cash flows**.

The net present value of a sequence of cash flows received at different points in tim

	A	B	C	D
1				
2		**Interest Rate**	0.2	
3			**Cash Flows**	
4		**Time**	**Investment 1**	**Investment 2**
5		1	-10000	-5000
6		2	25000	20000
7		3	-7000	-8000
8		**Total**	8000	7000

At face value, Investment 1 looks better than Investment 2. However, you can decide on which investment is better only when you know the true worth of the investment as of today. You can use the NPV function to calculate the returns.

The cash flows can occur

At the end of every year.

At the beginning of every year.

In the middle of every year.

NPV function assumes that the cash flows are at the end of the year. If the cash flows occur at different

	A	B	C	D
1				
2		**Interest Rate**	0.2	
3			**Cash Flows**	
4		**Time**	**Investment 1**	**Investment 2**
5		1	-10000	-5000
6		2	25000	20000
7		3	-7000	-8000
8		**Total**	=SUM(C5:C7)	=SUM(D5:D7)
9				
10		**NPV (End Year)**	=NPV(C2,C5:C7)	=NPV(C2,D5:D7)

You will get the following results −

	A	B	C	D
1				
2		**Interest Rate**	0.2	
3			**Cash Flows**	
4		**Time**	**Investment 1**	**Investment 2**
5		1	-10000	-5000
6		2	25000	20000
7		3	-7000	-8000
8		**Total**	8000	7000
9				
10		**NPV (End Year)**	4,976.85	5,092.59

As you observe NPV for Investment 2 is higher than that for Investment 1. Hence, Investment 2 is a better choice.

You got this result as cash out flows for Investment 2 are at later periods as compared to that of Investment 1

Cash Flows at the Beginning of the Year

Suppose the cash flows occur at the beginning of every year. In such a case, you should not include the first cash flow in NPV calculation as it already represents the current value.

You need to add the first cash flow to the NPV obtained from rest of the cash flows to get the net present value.

	A	B	C	D
1				
2		**Interest Rate**	0.2	
3			**Cash Flows**	
4		**Time**	**Investment 1**	**Investment 2**
5		1	-10000	-5000
6		2	25000	20000
7		3	-7000	-8000
8		**Total**	=SUM(C5:C7)	=SUM(D5:D7)
9				
10		**NPV (End Year)**	=NPV(C2,C5:C7)	=NPV(C2,D5:D7)
11				
12		**NPV (Beg. Year)**	=C5+NPV(C2,C6:C7)	=D5+NPV(C2,D6:D7)

You will get the following results −

	A	B	C	D
1				
2		**Interest Rate**	0.2	
3			**Cash Flows**	
4		**Time**	**Investment 1**	**Investment 2**
5		1	-10000	-5000
6		2	25000	20000
7		3	-7000	-8000
8		**Total**	8000	7000
9				
10		**NPV (End Year)**	4,976.85	5,092.59
11				
12		**NPV (Beg. Year)**	5,972.22	6,111.11

Cash Flows in the Middle of the Year

Suppose the cash flows occur in the middle of every year. In such a case, you need to multiply the NPV obtained from the cash flows by $\sqrt{1+r}$ to get the net present value.

	A	B	C	D
1				
2		**Interest Rate**	0.2	
3			**Cash Flows**	
4		**Time**	**Investment 1**	**Investment 2**
5		1	-10000	-5000
6		2	25000	20000
7		3	-7000	-8000
8		**Total**	=SUM(C5:C7)	=SUM(D5:D7)
9				
10		**NPV (End Year)**	=NPV(C2,C5:C7)	=NPV(C2,D5:D7)
11				
12		**NPV (Beg. Year)**	=C5+NPV(C2,C6:C7)	=D5+NPV(C2,D6:D7)
13				
14		**NPV (Middle Year)**	=SQRT(1+C2)*C10	=SQRT(1+C2)*D10

You will get the following results :-

	A	B	C	D
1				
2		**Interest Rate**	0.2	
3			**Cash Flows**	
4		**Time**	**Investment 1**	**Investment 2**
5		1	-10000	-5000
6		2	25000	20000
7		3	-7000	-8000
8		**Total**	8000	7000
9				
10		**NPV (End Year)**	4,976.85	5,092.59
11				
12		**NPV (Beg. Year)**	5,972.22	6,111.11
13				
14		**NPV (Middle Year)**	5,451.87	5,578.66

Cash Flows at Irregular Intervals

If you want to calculate the net present value with irregular cash flows, i.e. cash flows occurring at random times, the calculation is a bit complex.

However, in Excel, you can easily do such a calculation with XNPV function.

Arrange your data with the dates and the cash flows.

Note − The first date in your data should be the earliest of all the dates. The other dates can occur in any order.

Use the XNPV function to calculate the net present value.

You will get the following results −

	A	B	C
1			
2		**Interest Rate**	0.2
3		**Date**	**Cash flows**
4		42536	5000
5		42657	5143
6		42855	8838
7		42684	-4893
8		42629	-2134
9		42843	8047
10		42609	3908
11		42568	-4007
12			
13		**Net Present Value**	=XNPV(C2,C4:C11,B4:B11)

Suppose today's date is 15th March, 2015. As you observe, all the dates of cash flows are of later dates.

If you want to find the net present value as of today, include it in the data at the top and specify 0 for the cash flow.

	A	B	C
1			
2		**Interest Rate**	0.2
3		**Date**	**Cash flows**
4		6/15/2016	5000
5		10/14/2016	5143
6		4/30/2017	8838
7		11/10/2016	-4893
8		9/16/2016	-2134
9		4/18/2017	8047
10		8/27/2016	3908
11		7/17/2016	-4007
12			
13		**Net Present Value**	17523.65

You will get the following results −

	A	B	C
1			
2		**Interest Rate**	0.2
3		**Date**	**Cash flows**
4		42078	0
5		42536	5000
6		42657	5143
7		42855	8838
8		42684	-4893
9		42629	-2134
10		42843	8047
11		42609	3908
12		42568	-4007
13			
14		**Net Present Value**	=XNPV(C2,C4:C12,B4:B12)

Internal Rate of Return (IRR)

Internal Rate of Return (IRR) of an investment is the rate of interest at which NPV is 0. It is the rate value for which the present values of the positive cash flows exactly compensate the negative ones. When the discount rate is the IRR, the investment is perfectly indifferent, i.e. the investor is neither gaining nor losing money.

Consider the following cash flows, different interest rates and the corresponding NPV values.



	A	B	C
1			
2		**Interest Rate**	0.2
3		**Date**	**Cash flows**
4		3/15/2015	0
5		6/15/2016	5000
6		10/14/2016	5143
7		4/30/2017	8838
8		11/10/2016	-4893
9		9/16/2016	-2134
10		4/18/2017	8047
11		8/27/2016	3908
12		7/17/2016	-4007
13			
14		**Net Present Value**	13940.18

As you can observe between the values of interest rate 10% and 11%, the sign of NPV changes.

When you fine-tune the interest rate to 10.53%, NPV is nearly 0. Hence, IRR is 10.53%.

Determining IRR of Cash Flows for a Project

You can calculate IRR of cash flows with Excel function IRR.



	A	B
1		
2		Cash Flows
3		10000
4		-5000
5		-8500
6		2000
7		
8		IRR
9		=IRR(B3:B6)

	A	B
1		
2		Cash Flows
3		10000
4		-5000
5		-8500
6		2000
7		
8		IRR
9		10.53%



	A	B
1		
2		Cash Flows
3		10000
4		-5000
5		-8500
6		2000
7		
8		IRR
9		=IRR(B3:B6)

	A	B
1		
2		Cash Flows
3		10000
4		-5000
5		-8500
6		2000
7		
8		IRR
9		10.53%

The IRR is 10.53% as you had seen in the previous section.

For the given cash flows, IRR may −

exist and unique

exist and multiple

Unique IRR

If IRR exists and is unique, it can be used to choose the best investment among several possibilities.

If the first cash flow is negative, it means the investor has the money and wants to invest. Then, the higher the IRR the better, since it represents the interest rate the investor is receiving.

If the first cash flow is positive, it means the investor needs money and is looking for a loan, the lower the IRR the better since it represents the interest rate the investor is paying.

To find if an IRR is unique or not, vary the guess value and calculate IRR. If IRR remains constant then it is unique.

CoachX LIVE

	A	B	C	D	E
1					
2		**Cash Flows**		**Guess**	**IRR**
3		10000			=IRR(B3:B6)
4		-5000		0.05	=IRR($B$3:$B$6,D4)
5		-8500		0.15	=IRR($B$3:$B$6,D5)
6		2000		0.2	=IRR($B$3:$B$6,D6)
7				0.25	=IRR($B$3:$B$6,D7)
8				0.3	=IRR($B$3:$B$6,D8)
9				0.35	=IRR($B$3:$B$6,D9)
10				0.4	=IRR($B$3:$B$6,D10)
11				0.45	=IRR($B$3:$B$6,D11)
12				0.5	=IRR($B$3:$B$6,D12)
13				0.55	=IRR($B$3:$B$6,D13)

As you observe, the IRR has a unique value for the different guess values.



	A	B	C	D	E
1					
2		**Cash Flows**		**Guess**	**IRR**
3		10000			10.53%
4		-5000		5.00%	10.53%
5		-8500		15.00%	10.53%
6		2000		20.00%	10.53%
7				25.00%	10.53%
8				30.00%	10.53%
9				35.00%	10.53%
10				40.00%	10.53%
11				45.00%	10.53%
12				50.00%	10.53%
13				55.00%	10.53%

Multiple IRRs

In certain cases, you may have multiple IRRs. Consider the following cash flows. Calculate IRR with different guess values.



	A	B	C	D	E
1					
2		Cash Flows		Guess	IRR
3		-20000			=IRR(B3:B6)
4		82000		0.15	=IRR($B$3:$B$6,D4)
5		-60000		0.2	=IRR($B$3:$B$6,D5)
6		2000		0.25	=IRR($B$3:$B$6,D6)
7				0.3	=IRR($B$3:$B$6,D7)
8				0.35	=IRR($B$3:$B$6,D8)
9				0.4	=IRR($B$3:$B$6,D9)
10				0.45	=IRR($B$3:$B$6,D10)
11				0.5	=IRR($B$3:$B$6,D11)
12				0.55	=IRR($B$3:$B$6,D12)
13				0.6	=IRR($B$3:$B$6,D13)

You will get the following results −

	A	B	C	D	E
1					
2		Cash Flows		Guess	IRR
3		-20000			-9.59%
4		82000		15.00%	-9.59%
5		-60000		20.00%	-9.59%
6		2000		25.00%	-9.59%
7				30.00%	-9.59%
8				35.00%	-9.59%
9				40.00%	-9.59%
10				45.00%	216.09%
11				50.00%	216.09%
12				55.00%	216.09%
13				60.00%	216.09%

You can observe that there are two IRRs - -9.59% and 216.09%. You can verify these two IRRs calculating NPV.

	A	B	C	D	E
1					
2		Cash Flows		Guess	IRR
3		-20000			-9.59%
4		82000		15.00%	-9.59%
5		-60000		20.00%	-9.59%
6		2000		25.00%	-9.59%
7				30.00%	-9.59%
8				35.00%	-9.59%
9				40.00%	-9.59%
10				45.00%	216.09%
11				50.00%	216.09%
12				55.00%	216.09%
13				60.00%	216.09%
14					
15				IRR	NPV
16				-9.59%	($0.00)
17				216.09%	$0.00

For both -9.59% and 216.09%, NPV is 0.

No IRRs

In certain cases, you may not have IRR. Consider the following cash flows. Calculate IRR with different guess values.

	A	B	C	D	E
1					
2		**Cash Flows**		**Guess**	**IRR**
3		10000			=IRR(B3:B6)
4		-5000		0.05	=IRR($B$3:$B$6,D4)
5		8500		0.15	=IRR($B$3:$B$6,D5)
6		2000		0.2	=IRR($B$3:$B$6,D6)
7				0.25	=IRR($B$3:$B$6,D7)
8				0.3	=IRR($B$3:$B$6,D8)
9				0.35	=IRR($B$3:$B$6,D9)
10				0.4	=IRR($B$3:$B$6,D10)
11				0.45	=IRR($B$3:$B$6,D11)
12				0.5	=IRR($B$3:$B$6,D12)
13				0.55	=IRR($B$3:$B$6,D13)

You will get the result as #NUM for all the guess values.

	A	B	C	D	E
1					
2		Cash Flows		Guess	IRR
3		10000			#NUM!
4		-5000		5.00%	#NUM!
5		8500		15.00%	#NUM!
6		2000		20.00%	#NUM!
7				25.00%	#NUM!
8				30.00%	#NUM!
9				35.00%	#NUM!
10				40.00%	#NUM!
11				45.00%	#NUM!
12				50.00%	#NUM!
13				55.00%	#NUM!

The result #NUM means that there is no IRR for the cash flows considered.

	A	B	C	D	E
1					
2		Cash Flows		Guess	IRR
3		10000			#NUM!
4		-5000		5.00%	#NUM!
5		8500		15.00%	#NUM!
6		2000		20.00%	#NUM!
7				25.00%	#NUM!
8				30.00%	#NUM!
9				35.00%	#NUM!
10				40.00%	#NUM!
11				45.00%	#NUM!
12				50.00%	#NUM!
13				55.00%	#NUM!

The result #NUM means that there is no IRR for the cash flows considered.

Cash Flows Patterns and IRR

If there is only one sign change in the cash flows, such as from negative to positive or positive to negative, then a unique IRR is guaranteed. For example, in capital investments, the first cash flow will be negative, while the rest of the cash flows will be positive. In such cases, unique IRR exists.

If there is more than one sign change in the cash flows, IRR may not exist. Even if it exists, it may not be unique.

Decisions based on IRRs

Many analysts prefer to use IRR and it is a popular profitability measure because, as a percentage, it is easy to understand and easy to compare to the required return. However, there are certain problems while making decisions with IRR. If you rank with IRRs and make decisions based on these ranks, you may end up with wrong decisions.

You have already seen that NPV will enable you to make financial decisions. However, IRR and NPV will not always lead to the same decision when projects are mutually exclusive.

Mutually exclusive projects are those for which the selection of one project precludes the acceptance of another. When projects that are being compared are mutually exclusive, a ranking conflict may arise between NPV and IRR. If you have to choose between project A and project B, NPV may suggest acceptance of project A whereas IRR may suggest project B.

This type of conflict between NPV and IRR may arise because of one of the following reasons −

The projects are of greatly different sizes, or

The timing of the cash flows are different.

Projects of significant size difference

	A	B	C	D
1				
2			**Investment**	**IRR**
3		**Project A**	1000	10%
4		**Project B**	100	50%

If you want to make a decision by IRR, project A yields a return of 100 and Project B a return of 50. Hence, investment on project A looks profitable. However, this is a wrong decision because of the difference in the scale of projects.

Consider −

You have 1000 to invest.

If you invest entire 1000 on project A, you get a return of 100.

If you invest 100 on project B, you will still have 900 in your hand that you can invest on another project, say project C. Suppose you get a return of 20% on project C, then the total return on project B and project C is 230, which is way ahead in profitability.

Thus, NPV is a better way for decision making in such cases.

Projects with different cash flows timings

CoachX LIVE



	A	B	C	D
1				
2		**Year**	**Project A**	**Project B**
3		0	-1000	-1000
4		1	0	400
5		2	200	400
6		3	300	300
7		4	500	300
8		5	900	200
9		**IRR**	17%	20%
10		**NPV**	815.89	552.40

Again, if you consider IRR to decide, project B would be the choice. However, project A has a higher NPV and is an ideal choice.

IRR of Irregularly Spaced Cash Flows (XIRR)

Your cash flows may sometimes be irregularly spaced. In such a case, you cannot use IRR as IRR requires equally spaced time intervals. You can use XIRR instead, which takes into account the dates of the cash flows along with the cash flows.



	A	B	C
1			
2		**Date**	**Cash Flows**
3		4/8/2015	-10000
4		8/15/2015	4000
5		3/15/2016	3000
6		4/25/2016	5000
7			
8		**XIRR**	=XIRR(C3:C6,B3:B6)

The Internal Rate of Return that results in is 26.42%

XIRR	26.42%

Modified IRR (MIRR)

Consider a case when your finance rate is different from your reinvestment rate. If you calculate Internal Rate of Return with IRR, it assumes same rate for both finance and reinvestment. Further, you might also get multiple IRRs.

For example, consider the cash flows given below −



	A	B	C	D
1				
2		Finance Rate		10%
3		Reinvestment Rate		12%
4				
5			Year	Cash flows
6			0	-1.6
7			1	10
8			2	-10
9				
10			Discount Rate	NPV
11			10%	($0.70)
12			25%	$0.00
13			110%	$0.43
14			400%	($0.00)
15			500%	($0.04)

As you observe, NPV is 0 more than once, resulting in multiple IRRs.

Further, reinvestment rate is not taken into account. In such cases, you can use modified IRR (MIRR)



	B	C	D
2	Finance Rate		0.1
3	Reinvestment Rate		0.12
5		Year	Cash flows
6		0	-1.6
7		1	10
8		2	-10
10		Discount Rate	NPV
11		0.1	=NPV(C11,D6:D8)
12		0.25	=NPV(C12,D6:D8)
13		1.1	=NPV(C13,D6:D8)
14		4	=NPV(C14,D6:D8)
15		5	=NPV(C15,D6:D8)
17	MIRR	=MIRR(D6:D8,D2,D3)	

You will get a result of 7% as shown below −



	A	B	C	D
1				
2		Finance Rate		10%
3		Reinvestment Rate		12%
4				
5			Year	Cash flows
6			0	-1.6
7			1	10
8			2	-10
9				
10			Rate	NPV
11			10%	($0.70)
12			25%	$0.00
13			110%	$0.43
14			400%	($0.00)
15			500%	($0.04)
16				
17		MIRR	7%	

The project provided a comprehensive understanding of financial functions in Excel, enabling effective financial analysis and decision-making.

Submission Guidelines

Format: PowerPoint or PDF
Length: 1-20 slides.
Sections: Introduction, Key Findings, Actionable, Methodologies , Approaches, Insights, Conclusions

Tools and Technologies :

EXCEL , WORD, PPT

Deadline:

Submit your report and presentation within 21 Days from the day you will start.

********** THANK YOU **********